

The *Circus* solution for the Steam Boiler Problem
(Corrected based on Jim Woodcock's original report)

Kangfeng Ye

July 20, 2016

Abstract

This is a *Circus* solution for the steam boiler control system problem. The specification is based on the original report back to 2002 by Jim Woodcock. Then we use the model checker to find errors and correct them afterwards. Therefore, it is the parsed, type-checked and model-checked version of *Circus* solutions for the steam boiler problem. But by now, it is not completely model-checked, such as deadlock free, livelock free, and other properties due to the state space explosion problem.

Contents

1	Introduction	2
1.1	Notes	2
2	The <i>Timer</i>	3
3	The <i>Analyser</i>	4
3.1	Parameters	4
3.2	Sensor	4
3.3	Pump	5
3.4	Valve	6
3.5	Expected values	7
3.6	Failures and repairs	8
3.6.1	Repair Failed Equipments	10
3.7	Input messages	12
3.8	Analysing messages	12
3.9	The <i>Analyser</i>	15
4	The <i>Controller</i>	18
4.1	The formal paragraphs	18
5	The <i>Reporter</i>	21
6	Steam Boiler	24

Chapter 1

Introduction

This case study is based on the *Circus* solution for the steam boiler control system problem [1] from Jim Woodcock’s original technical report [3]. Additionally, I also read Leo Freitas’s parsable steam boiler [2] which is based on Jim Woodcock’s original version as well. The purpose of this work is to formalise the solution by the model checking approach [4] we have proposed recently. It is worth noting that this document omits most of description of this model in the original version for brevity. Therefore, it can be better understood with references to the original document.

The steps to apply our approach to this case are listed below.

- Step 1. Use *Circus2ZCSP* translator to link this specification to the combination of CSP and Z—consequently two files named *steam_boiler_z.tex* and *steam_boiler_csp.csp* respectively.
- Step 2. Load the two files into modified ProB.
- Step 3. Then use ProB’s model checking and animation functions to find errors. For errors, we modify this model to correct the problems, and then go back to “Step 1” again.

1.1 Notes

- We rename all identifies with subscript digits to underline (_) symbol. For example, M_1 to M_1 . That is due to the fact that subscript is not supported in CSP.
- According to Leo’s version, for the appropriate typesetting of the expected text in Unicode, the freetype should be given a \LaTeX markup directives and a \LaTeX command. For brevity, this model omits this additional \LaTeX definitions.

Chapter 2

The *Timer*

The header of a *Circus* model must include *circus_toolkit* as its parents.

```
section SteamBoiler parents circus_toolkit
```

```
channel clocktick, startcycle
```

In the original model, the *time* is initialised to *cyclelimit* by an assignment *time* := *cyclelimit*. In this model, we modify it to a schema expression (*InitTimer*). They are semantically equal. The reason of this modification is because, with this schema, in the final resultant *CSP* \parallel *Z* model, *time* is initialised in the early stage (during “initialisation” of the model) instead of in the later stage by the linked assignment in CSP. This will make the model checker easier to find the initial state.

The mod operator binds more tightly than + operator (albeit, it is not the case in mathematics), thus

```
( time := time + 1 mod cycletime )
```

will not get the expected result. It is corrected by adding additional brackets.

```
process Timer  $\hat{=}$  begin
  cycletime == 5
  cyclelimit == cycletime - 1
  Time == 0 .. cyclelimit
  state TimeState == [ time : Time ]
  InitTimer == [ TimeState ' | time ' = cyclelimit ]
  TimeOp == [  $\Delta$  TimeState | time '  $\geq$  time ]
  TCycle  $\hat{=}$  ( time := (time + 1) mod cycletime );
    ( if time = 0  $\longrightarrow$  startcycle  $\rightarrow$  Skip  $\parallel$  time  $\neq$  0  $\longrightarrow$  Skip fi );
    clocktick  $\rightarrow$  TCycle
  • ( InitTimer ) ; TCycle
end
```

Chapter 3

The *Analyser*

3.1 Parameters

MAX_NUM and $NUMS$ are introduced just for facilitating the animation.

| $MAX_NUM : \mathbb{N}$

$NUMS == 0 \dots MAX_NUM$

| $C, P, U_1, U_2, W : NUMS$

$M_1, N_1, N_2, M_2 : NUMS$
$M_1 \leq N_1 \leq N_2 \leq M_2$

3.2 Sensor

$Unit[X] == [a_1, a_2 : NUMS ; st : X \mid a_1 \leq a_2]$

$SState ::= sokay \mid sfailed$

$QSensor == Unit[SState][qa_1/a_1, qa_2/a_2, qst/st]$

$InitQSensor == [QSensor' \mid qa_1' = 0 \wedge qa_2' = C \wedge qst' = sokay]$

$VSensor == Unit[SState][va_1/a_1, va_2/a_2, vst/st]$

$InitVSensor == [VSensor' \mid va_1' = 0 \wedge va_2' = 0 \wedge vst' = sokay]$

3.3 Pump

$PState ::= popen \mid pwaiting \mid pclosed \mid pfailed$

$Pump0$ is rewritten to give a small size set $\{0, P\}$ as pa 's type to ease model checking. Since the values of pa_1 and pa_2 are implied from the pump state and not the input value from environment, it is safe to reduce the size of their type.

$Pump0 == [pa_1, pa_2 : \{0, P\} ; pst : PState \mid pa_1 \leq pa_2]$

$PumpOpen == [Pump0 \mid pst = popen \Rightarrow (pa_1 = P \wedge pa_2 = P)]$

$PumpWaitingOrClosed == [Pump0 \mid$
 $(pst = pwaiting \vee pst = pclosed) \Rightarrow (pa_1 = 0 \wedge pa_2 = 0)]$

$Pump == PumpOpen \wedge PumpWaitingOrClosed$

$InitPump == [PumpWaitingOrClosed' \mid pst' = pclosed]$

$PCState ::= pcflow \mid pcnoflow \mid pcfailed$
 $PumpCtr0 == [Pump ; pcst : PCState]$

$POpenPCFlowOrFailed == [PumpCtr0 \mid$
 $pst = popen \Rightarrow (pcst = pcflow \vee pcst = pcfailed)]$

$PWaitingPCNoFlowOrFailed == [PumpCtr0 \mid$
 $pst = pwaiting \Rightarrow (pcst = pcnoflow \vee pcst = pcfailed)]$

$PClosedPCNoFlowOrFailed == [PumpCtr0 \mid$
 $pst = pclosed \Rightarrow (pcst = pcnoflow \vee pcst = pcfailed)]$

$PFailedPCFlow == [PumpCtr0 \mid$
 $(pst = pfailed \wedge pcst = pcflow) \Rightarrow (pa_1 = P \wedge pa_2 = P)]$

$PFailedPCNoFlow == [PumpCtr0 \mid$
 $(pst = pfailed \wedge pcst = pcnoflow) \Rightarrow (pa_1 = 0 \wedge pa_2 = 0)]$

$PFailedPCFailed == [PumpCtr0 \mid$
 $(pst = pfailed \wedge pcst = pcfailed) \Rightarrow (pa_1 = 0 \wedge pa_2 = P)]$

$$\begin{aligned}
PumpCtr == & \\
& POpenPCFlowOrFailed \wedge PWaitingPCNoFlowOrFailed \wedge \\
& PClosedPCNoFlowOrFailed \wedge PFailedPCFlow \wedge PFailedPCNoFlow \wedge \\
& PFailedPCFailed
\end{aligned}$$

$$InitPumpCtr == [PumpCtr' \mid InitPump \wedge pcst' = pcnoflow]$$

$$PumpIndex == 1 \dots 4$$

The names of pa_1 and pa_2 are changed to pta_1 and pta_2 to avoid confusion. And their types are changed as well due to the same reason as pa_1 and pa_2 in $Pump0$.

$ \begin{aligned} & PumpCtrSystem \\ & pumpctr : PumpIndex \rightarrow PumpCtr \\ & pta_1, pta_2 : \{0, P, 2 * P, 3 * P, 4 * P\} \\ & pta_1 = (pumpctr\ 1).pa_1 + (pumpctr\ 2).pa_1 + \\ & \quad (pumpctr\ 3).pa_1 + (pumpctr\ 4).pa_1 \\ & pta_2 = (pumpctr\ 1).pa_2 + (pumpctr\ 2).pa_2 + \\ & \quad (pumpctr\ 3).pa_2 + (pumpctr\ 4).pa_2 \end{aligned} $
--

$ \begin{aligned} & InitPumpCtrSystem \\ & PumpCtrSystem' \\ & \exists InitPumpCtr \bullet \\ & \quad \forall i : PumpIndex \bullet pumpctr' i = \theta PumpCtr' \end{aligned} $
--

3.4 Valve

A freetype $VAction$ and a schema $SetValveState$ are added to update valve's state according to the output signal sent to the physical units. If this program sends $openValve$ (or $closeValve$), then its action is $openv$ (or $closev$) and its state should be $vopen$ (or $vclosed$). Otherwise, if none of $openValve$ and $closeValve$ is issued, then it is $VNoChange$ and its state is unchanged.

$$\begin{aligned}
VState &::= vopen \mid vclosed \\
VAction &::= openv \mid closev \mid VNoChange \\
Valve &== [valve : VState] \\
InitValve &== [Valve' \mid valve' = vclosed] \\
SetValveState &== [\Delta Valve ; vstate? : VAction \mid \\
& \quad (vstate? = VNoChange \Rightarrow valve' = valve) \wedge \\
& \quad (vstate? = openv \Rightarrow valve' = vopen) \wedge \\
& \quad (vstate? = closev \Rightarrow valve' = vclosed)]
\end{aligned}$$

3.5 Expected values

$CValues == [qc_1, qc_2, vc_1, vc_2 : NUMS]$

$InitCValues == [CValues' \mid qc_1' = 0 \wedge qc_2' = C \wedge vc_1' = 0 \wedge vc_2' = W]$

$QLowerBoundValveOpen == [CValues; Valve \mid valve = vopen \wedge qc_1 = 0]$

$QLowerBoundValveClosed ==$
 $[CValues; QSensor; VSensor; PumpCtrSystem; Valve \mid valve = vclosed \wedge$
 $qc_1 = \max\{0, qa_1 - 5 * va_2 - 12 * U_1 + 5 * pta_1\}]$

qc_2 must be larger than or equal to 0.

$QUpperBound ==$
 $[CValues; QSensor; VSensor; PumpCtrSystem \mid$
 $qc_2 = \max\{0, \min\{C, qa_2 - 5 * va_1 + 12 * U_2 + 5 * pta_2\}\}]$

$VLowerBound == [CValues; VSensor \mid vc_1 = \max\{0, va_1 - 5 * U_2\}]$

$vc_2 = \min\{W, va_2 - 5 * U_1\}$ should be $vc_2 = \min\{W, va_2 + 5 * U_1\}$.

$VUpperBound == [CValues; VSensor \mid (vc_2 = \min\{W, va_2 + 5 * U_1\})]$

$InputPState == \{popen, pclosed\}$

$InputPCState == \{pcflow, pcnoflow\}$

$ExpectedPumpStates$ $expectedp : PumpIndex \rightarrow InputPState$ $expectedpc : PumpIndex \rightarrow InputPCState$
--

We add a schema $InitExpectedPumpStates$ to initialise the expected pump states though their initial states can be arbitrarily chosen. In addition, we use abnormal combination of the pump state $pclosed$ and the pump controller state $pcflow$ to indicate this initial value should not be used to check again input pump and pump controller states.

$InitExpectedPumpStates$ $ExpectedPumpStates'$ $expectedp' = \{1 \mapsto pclosed, 2 \mapsto pclosed, 3 \mapsto pclosed, 4 \mapsto pclosed\}$ $expectedpc' = \{1 \mapsto pcflow, 2 \mapsto pcflow, 3 \mapsto pcflow, 4 \mapsto pcflow\}$
--

This schema $CalcExpectedPumpState$ is added to update expected pump and pump controller states according to output pump states to the physical units. If the output pump state is $popen$, then the

expected pump state is *popen* as well and the pump controller state will be *pcflow*. Otherwise, *pclosed* and *pcnoflow* respectively. At the same time, the pump state is changed to *pwaiting* in case the pump is expected to be opened from closed.

$ \begin{array}{l} \text{CalcExpectedPumpState} \text{ ---} \\ \Delta \text{ExpectedPumpStates} \\ \Delta \text{PumpCtrSystem} \\ \text{pumpstate?} : \text{PumpIndex} \rightarrow \text{InputPState} \\ \hline \forall i : \text{PumpIndex} \bullet \\ (\\ (\\ (\text{expectedp}' i = \text{pumpstate? } i) \wedge \\ (\\ (\text{pumpstate? } i = \text{popen} \wedge \text{expectedpc}' i = \text{pcflow}) \vee \\ (\text{pumpstate? } i = \text{pclosed} \wedge \text{expectedpc}' i = \text{pcnoflow}) \\) \\) \wedge \\ (\\ ((\text{pumpctr}' i).pst = \\ \text{if}(\text{expectedp } i = \text{pclosed} \wedge \\ \text{pumpstate? } i = \text{popen} \wedge \\ (\text{pumpctr } i).pst = \text{pclosed}) \\ \text{then} \\ \text{pwaiting} \\ \text{else} \\ (\text{pumpctr } i).pst \\) \wedge \\ (\text{pumpctr}' i).pcst = (\text{pumpctr } i).pcst \\) \\) \end{array} $
--

$$\begin{array}{l}
\text{Equipment0} == \\
QSensor \wedge VSensor \wedge \text{PumpCtrSystem} \wedge \text{Valve} \wedge \\
CValues \wedge \text{ExpectedPumpStates}
\end{array}$$

3.6 Failures and repairs

$$\begin{array}{l}
QFailed == [QSensor \mid qst = sfailed] \\
VFailed == [VSensor \mid vst = sfailed] \\
PFailed == [PumpCtrSystem \mid \\
(\exists i : \text{PumpIndex} \bullet (\text{pumpctr } i).pst = pfailed)] \\
PCFailed == [PumpCtrSystem \mid \\
(\exists i : \text{PumpIndex} \bullet (\text{pumpctr } i).pcst = pcfailed)] \\
\\
UnitFailure ::= qfail \mid vfail \mid pfail \langle \langle \text{PumpIndex} \rangle \rangle \mid pcfail \langle \langle \text{PumpIndex} \rangle \rangle \\
Failures == [failures, noacks : \mathbf{P} \text{ UnitFailure} \mid noacks \subseteq failures]
\end{array}$$

The original schema uses

$$(u = pfail \ i \wedge PFailed)$$

to calculate pump failures. However, since $PFailed$ holds if at least one of pumps is failed, the schema results in pump failures for all pumps. Finally, the schema is updated to check pump failures against individual pump state directly by

$$(u = pfail\ i \wedge (pumpctr\ i).pst = pfailed)$$

. This is the same case as $pcfail$.

<i>EquipmentFailures</i>	
<i>Equipment0</i>	
<i>Failures</i>	
<i>failures</i> =	
	$\{ u : UnitFailure ; i : PumpIndex \mid$ $(u = qfail \wedge QFailed) \vee$ $(u = vfail \wedge VFailed) \vee$ $(u = pfail\ i \wedge (pumpctr\ i).pst = pfailed) \vee$ $(u = pcfail\ i \wedge (pumpctr\ i).pcst = pcfailed)$ $\bullet u \}$

$$InitFailures == [Failures' \mid failures' = \emptyset \wedge noacks' = \emptyset]$$

$$FailuresExpected == [Failures ; failureacks : \mathbf{P}\ UnitFailure \mid failureacks \subseteq noacks]$$

$$AcceptFailureAcks == [\Delta Failures ; FailuresExpected \mid noacks' = noacks \setminus failureacks]$$

$$RepairsExpected == [Failures ; repairs : \mathbf{P}\ UnitFailure \mid repairs \subseteq failures]$$

$$AcceptRepairs == [\Delta Failures ; RepairsExpected \mid failures' = failures \setminus repairs \wedge noacks' = noacks \setminus repairs]$$

The schema *UpdateFailuresAck* is added to update *noacks* according to input *failureacks?* and *repairs?*.

- For the new failures identified in this cycle, we add them to *noacks* to state they are not acknowledged.
- If *failureacks?* is accepted, that is $failureacks? \subseteq noacks$, we take these acknowledged failures out of *noacks*.
- If *repairs?* is accepted, that is $repairs? \subseteq failures$, we take these repaired failures out of *noacks*.

UpdateFailuresAck

$\Delta Failures$

failureacks? : \mathbf{P} *UnitFailure*

repairs? : \mathbf{P} *UnitFailure*

$\exists newnoacks : \mathbf{P} \text{ UnitFailure} \bullet ($
 $(newnoacks = noacks \cup (failures' \setminus failures)) \wedge$
 $($
 $((failureacks? \subseteq noacks) \wedge (repairs? \subseteq failures))$
 $\Rightarrow (noacks' = newnoacks \setminus (failureacks? \cup repairs?))) \wedge$
 $((failureacks? \subseteq noacks) \wedge \neg(repairs? \subseteq failures))$
 $\Rightarrow (noacks' = newnoacks \setminus failureacks?) \wedge$
 $((\neg(failureacks? \subseteq noacks) \wedge (repairs? \subseteq failures))$
 $\Rightarrow (noacks' = newnoacks \setminus repairs?) \wedge$
 $((\neg(failureacks? \subseteq noacks) \wedge \neg(repairs? \subseteq failures))$
 $\Rightarrow (noacks' = newnoacks))$
 $)$
 $)$

$Equipment == (QLowerBoundValveOpen \vee QLowerBoundValveClosed) \wedge$
 $QUpperBound \wedge VLowerBound \wedge VUpperBound \wedge$
 $ExpectedPumpStates \wedge EquipmentFailures$

In *InitEquipment*, expected pump and pump controller states and valve state are initialised as well.

$InitEquipment == Equipment0' \wedge InitQSensor \wedge InitVSensor \wedge$
 $InitPumpCtrSystem \wedge InitCValues \wedge InitFailures \wedge$
 $InitExpectedPumpStates \wedge InitValve$

3.6.1 Repair Failed Equipments

This is a newly added section to repair equipments according to input *repairs?*.

For *QSensor*, if it is repaired, then its *qst* will be *sokay*. Otherwise it stays unchanged.

RepairQSensor

$\Delta QSensor$

repairs? : \mathbf{P} *UnitFailure*

$qa_1' = qa_1$
 $qa_2' = qa_2$
 $qfail \in repairs? \Rightarrow qst' = okay$
 $qfail \notin repairs? \Rightarrow qst' = qst$

For *VSensor*, if it is repaired, then its *vst* will be *sokay*. Otherwise it stays unchanged.

RepairVSensor

$\Delta VSensor$

repairs? : \mathbf{P} *UnitFailure*

$va_1' = va_1$
 $va_2' = va_2$
 $vfail \in repairs? \Rightarrow vst' = okay$
 $vfail \notin repairs? \Rightarrow vst' = vst$

If a pump controller is repaired, its state will be *pcflow* if current pump state is *popen*, or its state will be *pcnoflow* if current pump state is not *popen*.

$RepairAPumpCtr$	$\Delta PumpCtr$
$pst' = pst$	
$pst = popen \Rightarrow pcst' = pcflow$	
$pst \neq popen \Rightarrow pcst' = pcnoflow$	

If a pump is repaired, its state will be *pclosed* and its pump controller state stays unchanged.

$RepairAPump$	$\Delta PumpCtr$
$pst' = pclosed$	
$pcst' = pcst$	

If both a pump and its controller are repaired, then the pump will be *pclosed* and its controller will be *pcnoflow*.

$RepairPumpCtrAndPump$	$\Delta PumpCtr$
$pst' = pclosed$	
$pcst' = pcnoflow$	

The schema *RepairPumps* repairs all pumps and their controllers according to input *repairs?*.

$RepairPumps$	$\Delta PumpCtrSystem$
$repairs? : \mathbf{P} \text{ UnitFailure}$	
$\forall i : PumpIndex \bullet$	
$\exists PumpCtr ; PumpCtr' \bullet$	
$(\theta PumpCtr' = pumpctr' i) \wedge (\theta PumpCtr = pumpctr i) \wedge$	
$((pfail i \in repairs? \wedge pcfail i \notin repairs?)$	
$\Rightarrow RepairAPump) \wedge$	
$((pfail i \notin repairs? \wedge pcfail i \in repairs?)$	
$\Rightarrow RepairAPumpCtr) \wedge$	
$((pfail i \in repairs? \wedge pcfail i \in repairs?)$	
$\Rightarrow RepairPumpCtrAndPump) \wedge$	
$((pfail i \notin repairs? \wedge pcfail i \notin repairs?)$	
$\Rightarrow \theta PumpCtr' = \theta PumpCtr)$	
$)$	

The *RepairEquipments* tries to repair all equipments according to input *repairs?*. If *repairs?* are accepted, all equipments will be repaired. Otherwise, all equipments will stay unchanged.

$$\begin{aligned}
RepairEquipments == & \\
& (RepairsExpected[repairs? / repairs] \wedge \\
& \quad RepairPumps \wedge RepairQSensor \wedge RepairVSensor \\
&) \vee \\
& ((\neg RepairsExpected[repairs? / repairs]) \wedge \\
& \quad \exists PumpCtrSystem \wedge \exists QSensor \wedge \exists VSensor \\
&)
\end{aligned}$$

A *emergencyCond* state is introduced to indicate if both input *repairs?* and *failureacks?* are accepted or not. It is set to 1 if there is unaccepted *repairs?* or *failureacks?*, or both. Otherwise, it is set to 0.

This update happens in the beginning of each cycle and the value is used in the later of the cycle.

$$\begin{aligned}
& \text{EmergencyCond} == [\text{emergencyCond} : \{0, 1\}] \\
& \text{MarkEmergencyCond} == [\Delta \text{EmergencyCond} \mid \text{emergencyCond}' = 1] \\
& \text{ClearEmergencyCond} == [\Delta \text{EmergencyCond} \mid \text{emergencyCond}' = 0] \\
& \text{EvalRepairFailureAck} == \\
& \quad (\text{RepairsExpected}[\text{repairs?} / \text{repairs}] \wedge \\
& \quad \quad \text{FailuresExpected}[\text{failureacks?} / \text{failureacks}] \wedge \\
& \quad \quad \text{ClearEmergencyCond} \\
& \quad) \vee \\
& \quad ((\neg \text{RepairsExpected}[\text{repairs?} / \text{repairs}] \vee \\
& \quad \quad \neg \text{FailuresExpected}[\text{failureacks?} / \text{failureacks}]) \\
& \quad \quad \wedge \text{MarkEmergencyCond} \\
& \quad)
\end{aligned}$$

3.7 Input messages

$$\begin{aligned}
& \text{InputSignal} ::= \\
& \quad \text{stop} \mid \text{steamBoilerWaiting} \mid \text{physicalUnitsReady} \mid \text{transmissionFailure}
\end{aligned}$$

$$\begin{aligned}
& \text{UnitState} \text{ ---} \\
& \text{pumpState} : \text{PumpIndex} \rightarrow \text{InputPState} \\
& \text{pumpCtrState} : \text{PumpIndex} \rightarrow \text{InputPCState} \\
& q, v : \text{NUMS}
\end{aligned}$$

$$\begin{aligned}
& \text{InputMsg} \text{ ---} \\
& \text{signals} : \mathbf{P} \text{InputSignal} \\
& \text{UnitState} \\
& \text{failureacks, repairs} : \mathbf{P} \text{UnitFailure}
\end{aligned}$$

3.8 Analysing messages

The input value $x?$ should be checked against calculated values c_1 and c_2 , instead of adjusted values a_1 and a_2 .

$$\begin{aligned}
& \text{Expected} == [x?, c_1, c_2 : \text{NUMS} \mid c_1 \leq x? \leq c_2] \\
& \text{Unexpected} == \neg \text{Expected}
\end{aligned}$$

$$\text{Sensor} == [\Delta \text{Unit}[S\text{State}] ; c_1, c_2, c_1', c_2', x? : \text{NUMS}]$$

$$\begin{aligned}
& \text{CheckAndAdjustSensor} \text{ ---} \\
& \text{Sensor} \text{ ---} \\
& \quad \text{Expected} \Rightarrow st' = st \\
& \quad \text{Unexpected} \Rightarrow st' = \text{sfailed} \\
& \quad st' = \text{sokay} \Rightarrow a_1' = x? \wedge a_2' = x? \\
& \quad st' = \text{sfailed} \Rightarrow a_1' = c_1 \wedge a_2' = c_2
\end{aligned}$$

$$\begin{aligned}
& \text{CheckAndAdjustQ} == \text{QSensor} \wedge \\
& \quad \text{CheckAndAdjustSensor}[\\
& \quad \quad q?/x?, qa_1/a_1, qa_2/a_2, qc_1/c_1, qc_2/c_2, qst/st, \\
& \quad \quad qa_1'/a_1', qa_2'/a_2', qc_1'/c_1', qc_2'/c_2', qst'/st'] \\
& \text{CheckAndAdjustV} == \text{VSensor} \wedge \\
& \quad \text{CheckAndAdjustSensor}[\\
& \quad \quad v?/x?, va_1/a_1, va_2/a_2, vc_1/c_1, vc_2/c_2, vst/st, \\
& \quad \quad va_1'/a_1', va_2'/a_2', vc_1'/c_1', vc_2'/c_2', vst'/st']
\end{aligned}$$

The *ExpectedPumpStateTBD* checks if the expected pumps and their controllers state are undetermined. This happens in the initialisation stage when the expected states are unknown. And we indicate this in *InitExpectedPumpStates*.

$ \begin{aligned} & \text{ExpectedPumpStateTBD} \\ & \quad \text{exppst} : \text{InputPState} \\ & \quad \text{exppcst} : \text{InputPCState} \end{aligned} $
$ \begin{aligned} & \text{exppst} = \text{pclosed} \\ & \text{exppcst} = \text{pcflow} \end{aligned} $

If expected pump states are unknown, we adjust pumps and their controllers states according to input states only and will not check expected pump states.

$ \begin{aligned} & \text{CheckAndAdjustPumpTBD} \\ & \quad \Delta \text{PumpCtr} \\ & \quad \text{pst?}, \text{exppst} : \text{InputPState} \\ & \quad \text{pcst?}, \text{exppcst} : \text{InputPCState} \end{aligned} $
$ \begin{aligned} & ((\text{pst?} = \text{popen} \wedge \text{pcst?} = \text{pcflow}) \vee (\text{pst?} = \text{pclosed} \wedge \text{pcst?} = \text{pcnoflow})) \\ & \quad \Rightarrow (\text{pst}' = \text{pst?} \wedge \text{pcst}' = \text{pcst?}) \\ & (\text{pst?} = \text{popen} \wedge \text{pcst?} = \text{pcnoflow}) \Rightarrow (\text{pst}' = \text{pfailed} \wedge \text{pcst}' = \text{pcnoflow}) \\ & (\text{pst?} = \text{pclosed} \wedge \text{pcst?} = \text{pcflow}) \Rightarrow (\text{pst}' = \text{pfailed} \wedge \text{pcst}' = \text{pcflow}) \end{aligned} $

However, if expected pump states are valid, we adjust pumps and their controllers states according to input and expected pump states together.

$ \begin{aligned} & \text{CheckAndAdjustPump} \\ & \quad \Delta \text{PumpCtr} \\ & \quad \text{pst?}, \text{exppst} : \text{InputPState} \\ & \quad \text{pcst?}, \text{exppcst} : \text{InputPCState} \end{aligned} $
$ \begin{aligned} & ((\text{pst} = \text{pfailed} \wedge \text{pst}' = \text{pst}) \vee \\ & \quad (\text{pst} \neq \text{pfailed} \wedge \\ & \quad \quad (\text{pst?} = \text{exppst} \Rightarrow \text{pst}' = \text{pst?}) \wedge \\ & \quad \quad (\text{pst?} \neq \text{exppst} \Rightarrow \text{pst}' = \text{pfailed}) \\ & \quad) \\ &) \\ & ((\text{pcst} = \text{pcfailed} \wedge \text{pcst}' = \text{pcst}) \vee \\ & \quad (\text{pcst} \neq \text{pcfailed} \wedge \\ & \quad \quad (\text{pcst?} = \text{exppcst} \Rightarrow \text{pcst}' = \text{pcst?}) \wedge \\ & \quad \quad (\text{pcst?} \neq \text{exppcst} \Rightarrow \text{pcst}' = \text{pcfailed}) \\ & \quad) \\ &) \end{aligned} $

PromotePumpCheck

$\Delta PumpCtr$
 $\Delta PumpCtrSystem$
 $ExpectedPumpStates$
 $pst?, exppst : InputPState$
 $pcst?, exppcst : InputPCState$
 $pumpState? : PumpIndex \rightarrow InputPState$
 $pumpCtrState? : PumpIndex \rightarrow InputPCState$
 $i : PumpIndex$

$\theta PumpCtr = pumpctr\ i$
 $\theta PumpCtr' = pumpctr'\ i$
 $pst? = pumpState?\ i$
 $pcst? = pumpCtrState?\ i$
 $exppest = expectedp\ i$
 $exppcst = expectedpc\ i$

$SetPumpCtr == \forall i : PumpIndex \bullet$
 $\exists PumpCtr ; PumpCtr' ; pst?, exppst : PState ; pcst?, exppcst : PCState \bullet$
 $(PromotePumpCheck \wedge$
 $((CheckAndAdjustPumpTBD \wedge ExpectedPumpStateTBD) \vee$
 $(CheckAndAdjustPump \wedge \neg ExpectedPumpStateTBD))$
 $)$
 $)$

The original predicate of *StopPresent* has correct. Just because we introduce *NUMS* for animation, the predicate of *StopPresent* is modified too.

StopPresent

$signals? : \mathbf{P}\ InputSignal$
 $stops, stops' : NUMS$

$stop \in signals?$
 $((stops + 1 > MAX_NUM \wedge stops' = stops) \vee (stops' = stops + 1))$

StopNotPresent

$signals? : \mathbf{P}\ InputSignal$
 $stops, stops' : NUMS$

$stop \notin signals? \wedge stops < 3$
 $stops' = 0$

TooManyStops

$signals? : \mathbf{P}\ InputSignal$
 $stops, stops' : NUMS$

$stop \notin signals? \wedge stops \geq 3$
 $stops' = stops$

$AdjustStops == StopPresent \vee StopNotPresent \vee TooManyStops$

3.9 The Analyser

```

channel levelbelowmin, levelabovemax
channel emergencystop, cfailures, levelokay, nonqfailure :  $\mathbb{B}$ 
channel physicalunitsready, qfailure, sbwaiting, vzero :  $\mathbb{B}$ 

```

For animation purpose, *input* has been split into seven small channels: *input1*, *input2*, *input3*, *input4*, *input5*, *input6*, and *input7*.

```

channel input1 : (P InputSignal)
channel input2 : (PumpIndex  $\rightarrow$  InputPState)
channel input3 : (PumpIndex  $\rightarrow$  InputPCState)
channel input4 : (NUMS)
channel input5 : (NUMS)
channel input6 : (P UnitFailure)
channel input7 : (P UnitFailure)
channel startexec

```

```

channel failuresrepairs : (P UnitFailure)  $\times$  (P UnitFailure)

```

```

channel pumps : (PumpIndex  $\rightarrow$  InputPState)  $\times$  VAction
channelset Information ==
  { emergencystop, cfailures, levelabovemax, levelbelowmin, levelokay,
    nonqfailure, physicalunitsready, qfailure, sbwaiting, vzero }

```

```

process Analyser  $\hat{=}$  begin

```

```

  state AnalyserState == [ Equipment0 ; Failures ; InputMsg;
    stops : NUMS ; signalhistory : P InputSignal ; EmergencyCond ]

```

```

  StopSignalHis == [ stops : NUMS ; signalhistory : P InputSignal ]
  PumpOp ==  $\exists$  QSensor  $\wedge$   $\exists$  VSensor  $\wedge$   $\exists$  Valve  $\wedge$   $\exists$  CValues  $\wedge$ 
     $\exists$  Failures  $\wedge$   $\exists$  ExpectedPumpStates  $\wedge$   $\exists$  InputMsg  $\wedge$ 
     $\exists$  StopSignalHis  $\wedge$   $\exists$  EmergencyCond

```

For *InputMsg*, its initial value can be arbitrarily chosen and it will not have impacts on the behaviour of the program. To ease model checking, we set a specific initial value in *InitAnalyserState*.

```

  InitAnalyserState == [ AnalyserState' |
    InitEquipment  $\wedge$  stops' = 0  $\wedge$  signalhistory' =  $\emptyset$   $\wedge$ 
     $\theta$  InputMsg' = (let signals ==  $\emptyset$ [InputSignal];
      pumpState ==
        { 1  $\mapsto$  pclosed, 2  $\mapsto$  pclosed, 3  $\mapsto$  pclosed, 4  $\mapsto$  pclosed };
      pumpCtrState ==
        { 1  $\mapsto$  pcnoflow, 2  $\mapsto$  pcnoflow, 3  $\mapsto$  pcnoflow, 4  $\mapsto$  pcnoflow };
      q == 0 ; v == 0 ; failureacks ==  $\emptyset$ [UnitFailure];
      repairs ==  $\emptyset$ [UnitFailure]  $\bullet$ 
       $\theta$  InputMsg)
     $\wedge$  emergencyCond' = 0 ]

```

$Analyse ==$
 $[\Delta AnalyserState ; InputMsg ? \mid \theta InputMsg' = \theta InputMsg ? \wedge$
 $CheckAndAdjustQ \wedge CheckAndAdjustV \wedge AdjustStops \wedge$
 $signalhistory' = signalhistory \cup signals? \wedge$
 $UpdateFailuresAck \wedge \exists PumpCtrSystem \wedge \exists ExpectedPumpStates \wedge$
 $\exists Valve \wedge Equipment' \wedge \exists EmergencyCond]$

In its predicate, $N_1 < qa_2$ should be $N_2 < qa_2$.

$DangerZone == [AnalyserState \mid qa_1 \geq M_1 \wedge qa_2 \leq M_2$
 $\Rightarrow qa_1 < N_1 \wedge N_2 < qa_2]$

Instead of checking $\neg RepairsExpected \vee \neg FailuresExpected$, we check *emergencyCond*, because in the later stage, the *failures* and *noacks* have been updated and not original values. Therefore, it is wrong to check *repairs?* and *failureacks?* against updated *failures* and *noacks*.

$EmergencyStopCond == [AnalyserState \mid$
 $stops \geq 3 \vee DangerZone \vee emergencyCond = 1 \vee$
 $transmissionFailure \in signals]$

 $LevelBelowMin == [AnalyserState \mid M_1 \leq qa_1 < N_1 \wedge qa_2 \leq N_2]$
 $LevelAboveMax == [AnalyserState \mid N_1 \leq qa_1 \wedge N_2 < qa_2 \leq M_2]$
 $LevelInRange == [AnalyserState \mid N_1 \leq qa_1 \wedge qa_2 \leq N_2]$
 $RateZero == [VSensor \mid va_1 = 0 \wedge va_2 = 0]$
 $AllPhysicalUnitsOkay ==$
 $[AnalyserState \mid \neg QFailed \wedge \neg VFailed \wedge \neg PFailed \wedge \neg PCFailed]$
 $OtherPhysicalUnitsFail == \neg QFailed \wedge \neg AllPhysicalUnitsOkay$
 $SteamBoilerWaiting ==$
 $[AnalyserState \mid steamBoilerWaiting \in signalhistory]$
 $PhysicalUnitsReady ==$
 $[AnalyserState \mid physicalUnitsReady \in signalhistory]$

HandleRepair, as a schema expression, is added to repair equipments.

$HandleRepair == RepairEquipments \wedge EvalRepairFailureAck \wedge \exists CValues$
 $\wedge \exists Failures \wedge \exists InputMsg \wedge \exists StopSignalHis \wedge$
 $\exists Valve \wedge \exists ExpectedPumpStates$

$AnalyserCycle \hat{=} startcycle \rightarrow input1?signals \rightarrow input2?pumpState \rightarrow$
 $input3?pumpCtrState \rightarrow input4?q \rightarrow input5?v \rightarrow$
 $input6?failureacks \rightarrow input7?repairs \rightarrow$
 $((HandleRepair) ; (SetPumpCtr \wedge PumpOp));$
 $(Analyse) ; startexec \rightarrow InfoService$

$PumpOp2 == \exists QSensor \wedge \exists VSensor \wedge \exists CValues \wedge$
 $\exists Failures \wedge \exists InputMsg \wedge \exists StopSignalHis \wedge$
 $\exists EmergencyCond$
 $SetExpectedPumpState ==$
 $CalcExpectedPumpState \wedge SetValveState \wedge PumpOp2$

InfoService $\hat{=}$ (*OfferInformation* ; *InfoService*) \square
failuresrepairs ! *noacks* ! *repairs* \rightarrow *pumps* ? *pumpstate* ? *vstate* \rightarrow
(*SetExpectedPumpState*) ; *AnalyserCycle*
OfferInformation $\hat{=}$
emergencystop.EmergencyStopCond \rightarrow **Skip**
 \square
sbwaiting.SteamBoilerWaiting \rightarrow **Skip**
 \square
vzero.RateZero \rightarrow **Skip**
 \square
(*LevelBelowMin*) & *levelbelowmin* \rightarrow **Skip**
 \square
(*LevelAboveMax*) & *levelabovemax* \rightarrow **Skip**
 \square
levelokay.LevelInRange \rightarrow **Skip**
 \square
physicalunitsready.PhysicalUnitsReady \rightarrow **Skip**
 \square
cfailures.(\neg *AllPhysicalUnitsOkay*) \rightarrow **Skip**
 \square
qfailure.QFailed \rightarrow **Skip**
 \square
nongfailure.OtherPhysicalUnitsFail \rightarrow **Skip**

• (*InitAnalyserState*) ; *AnalyserCycle*

end

channelset *TAnalyserInterface* == { *startcycle* }
process *TAnalyser* $\hat{=}$
Timer [*TAnalyserInterface*] *Analyser* \ *TAnalyserInterface*

Chapter 4

The Controller

```
Mode ::= initialisation | normal | degraded | rescue | emergencyStop
Nonemergency == {initialisation, normal, degraded, rescue}
```

```
channel startpumps, stoppumps, openvalve, closevalve, sendprogreedy
channel reportmode : Mode
channel startreport, endreport
channelset Reports ==
    { startpumps, stoppumps, openvalve, closevalve, sendprogreedy }
channelset TAControllerInterface == { startexec } ∪ Information
```

4.1 The formal paragraphs

```
process Controller ≡ begin
```

```
state ModelState == [ mode : Mode ]
InitController == [ ModelState' | mode' = initialisation ]
EnterMode ≡ m : Mode • reportmode !m → mode := m
```

In *emergencyStop* mode, it is not necessary to adjust level *AdjustLevel* and just end report by *endreport*.

```
ControllerCycle ≡ startexec → startreport → NewModeAnalysis;
    ((mode ≠ emergencyStop) & AdjustLevel □
    (mode = emergencyStop) & Skip);
endreport → ControllerCycle
NewModeAnalysis ≡ emergencystop.True → EnterMode(emergencyStop)
    □ emergencystop.False → (
        (mode = initialisation) & InitModeAnalysis
        □ (mode = normal) & NormalModeAnalysis
        □ (mode = degraded) & DegradedModeAnalysis
        □ (mode = rescue) & RescueModeAnalysis
        □ ((mode ∉ Mode \ {emergencyStop})) & Skip
    )
```

```

InitModeAnalysis  $\hat{=}$ 
  sbwaiting.True  $\rightarrow$ 
    ( vzero.True  $\rightarrow$ 
      ( qfailure.False  $\rightarrow$ 
        ( physicalunitsready.True  $\rightarrow$ 
          ( levelokay.True  $\rightarrow$ 
            ( cfailures.False  $\rightarrow$  EnterMode (normal) )  $\square$ 
            cfailures.True  $\rightarrow$  EnterMode (degraded) )  $\square$ 
            levelokay.False  $\rightarrow$  EnterMode (emergencyStop) )  $\square$ 
          physicalunitsready.False  $\rightarrow$ 
            ( levelokay.True  $\rightarrow$ 
              sendprogready  $\rightarrow$  Skip  $\square$ 
              levelokay.False  $\rightarrow$  Skip ) )  $\square$ 
            qfailure.True  $\rightarrow$  EnterMode (emergencyStop) )  $\square$ 
          vzero.False  $\rightarrow$  EnterMode (emergencyStop) )  $\square$ 
        sbwaiting.False  $\rightarrow$  Skip
      NormalModeAnalysis  $\hat{=}$ 
        cfailures.False  $\rightarrow$  Skip  $\square$ 
        qfailure.True  $\rightarrow$  EnterMode (rescue)  $\square$ 
        nongfailure.True  $\rightarrow$  EnterMode (degraded)
      DegradedModeAnalysis  $\hat{=}$ 
        qfailure.False  $\rightarrow$ 
          ( cfailures.True  $\rightarrow$  Skip  $\square$ 
            cfailures.False  $\rightarrow$  EnterMode (normal) )
           $\square$  qfailure.True  $\rightarrow$  EnterMode (rescue)
      RescueModeAnalysis  $\hat{=}$ 
        qfailure.True  $\rightarrow$  Skip  $\square$ 
        qfailure.False  $\rightarrow$  (
          cfailures.False  $\rightarrow$  EnterMode (normal)
           $\square$  cfailures.True  $\rightarrow$  EnterMode (degraded) )
    )
  AdjustLevel  $\hat{=}$  levelbelowmin  $\rightarrow$  RaiseLevel  $\square$ 
    levelabovemax  $\rightarrow$  ReduceLevel  $\square$ 
    levelokay.True  $\rightarrow$  RetainLevel
  RaiseLevel  $\hat{=}$  StartPumps;
    if mode = initialisation  $\longrightarrow$  CloseValve
     $\llbracket$  mode  $\neq$  initialisation  $\longrightarrow$  Skip
    fi
  ReduceLevel  $\hat{=}$  StopPumps;
    if mode = initialisation  $\longrightarrow$  OpenValve
     $\llbracket$  mode  $\neq$  initialisation  $\longrightarrow$  Skip
    fi
  RetainLevel  $\hat{=}$  StopPumps;
    if mode = initialisation  $\longrightarrow$  CloseValve
     $\llbracket$  mode  $\neq$  initialisation  $\longrightarrow$  Skip
    fi
  StartPumps  $\hat{=}$  startpumps  $\rightarrow$  Skip
  StopPumps  $\hat{=}$  stoppumps  $\rightarrow$  Skip
  OpenValve  $\hat{=}$  openvalve  $\rightarrow$  Skip
  CloseValve  $\hat{=}$  closevalve  $\rightarrow$  Skip

```

• (InitController) ; ControllerCycle

end

process $TAController \hat{=}$
 $(TAnalyser \llbracket TAControllerInterface \rrbracket Controller) \setminus TAControllerInterface$

Chapter 5

The *Reporter*

```
OutputSignal ::= programReady | openValve | closeValve |  
                levelFailureDetection | steamFailureDetection |  
                levelRepairedAcknowledgement | steamRepairedAcknowledgement
```

```
OutputMsg  
  mode : Mode  
  signals : P OutputSignal  
  pumpState : PumpIndex → InputPState  
  pumpFailureDetection : P UnitFailure  
  pumpCtrFailureDetection : P UnitFailure  
  pumpRepairedAcknowledgement : P UnitFailure  
  pumpCtrRepairedAcknowledgement : P UnitFailure
```

Similar to the *input* channel, the *output* channel is split too.

```
channel output1 : Mode  
channel output2 : (P OutputSignal)  
channel output3 : (PumpIndex → InputPState)  
channel output4 : (P UnitFailure)  
channel output5 : (P UnitFailure)  
channel output6 : (P UnitFailure)  
channel output7 : (P UnitFailure)
```

```
process Reporter ≐ begin
```

```
  state ReporterState == [ OutputMsg ; valveSt : VAction | true ]
```

Similar to the *Timer* process and initial value of *InputMsg*, we initialise *OutputMsg* as well though its

initial value can be arbitrarily chosen.

```

InitReporter == [ ReporterState' | valveSt' = VNoChange ∧
  θ OutputMsg' =
    (let mode == initialisation ; signals == ∅[OutputSignal];
      pumpState == {1 ↦ pclosed, 2 ↦ pclosed,
        3 ↦ pclosed, 4 ↦ pclosed};
      pumpFailureDetection == ∅[UnitFailure];
      pumpCtrFailureDetection == ∅[UnitFailure];
      pumpRepairedAcknowledgement == ∅[UnitFailure];
      pumpCtrRepairedAcknowledgement == ∅[UnitFailure]
      • θ OutputMsg) ]

ReportService ≜ GatherReports ; ReportService □
reportmode.emergencyStop → mode := emergencyStop ; TidyUp □
TidyUp

```

This schema is used to update *OutputMsg* according to the inputs *noacks* and *repairs* from the *Analyser* process.

```

FailuresRepairs == [ ΔReporterState ; noacks? : (P UnitFailure);
  repairs? : (P UnitFailure) |
  (signals' = signals ∪
    (if(qfail ∈ noacks?) then {levelFailureDetection} else ∅) ∪
    (if(vfail ∈ noacks?) then {steamFailureDetection} else ∅) ∪
    (if(qfail ∈ repairs? then {levelRepairedAcknowledgement}
      else ∅) ∪
    (if(vfail ∈ repairs? then {steamRepairedAcknowledgement}
      else ∅)) ∧
  pumpFailureDetection' =
    noacks? ∩ {i : PumpIndex • pfail i} ∧
  pumpCtrFailureDetection' =
    noacks? ∩ {i : PumpIndex • pcfail i} ∧
  pumpRepairedAcknowledgement' =
    repairs? ∩ {i : PumpIndex • pfail i} ∧
  pumpCtrRepairedAcknowledgement' =
    repairs? ∩ {i : PumpIndex • pcfail i} ∧
  mode' = mode ∧ valveSt' = valveSt ∧ pumpState' = pumpState ]

```



```

TidyUp  $\hat{=}$  endreport  $\rightarrow$  failuresrepairs ?noacks?repairs  $\rightarrow$  (FailuresRepairs);
  output1!mode  $\rightarrow$  output2!signals  $\rightarrow$  output3!pumpState  $\rightarrow$ 
  output4!pumpFailureDetection  $\rightarrow$  output5!pumpCtrFailureDetection  $\rightarrow$ 
  output6!pumpRepairedAcknowledgement  $\rightarrow$ 
  output7!pumpCtrRepairedAcknowledgement  $\rightarrow$ 
  pumps!pumpState!valveSt  $\rightarrow$  Skip
GatherReports  $\hat{=}$   $\square$  m : Nonemergency  $\bullet$  reportmode.m  $\rightarrow$  mode := m
   $\square$ 
  sendprogreedy  $\rightarrow$  signals := signals  $\cup$  {programReady}
   $\square$ 
  startpumps  $\rightarrow$  pumpState := PumpIndex  $\times$  {popen}
   $\square$ 
  stoppumps  $\rightarrow$  pumpState := PumpIndex  $\times$  {pclosed}
   $\square$ 
  openvalve  $\rightarrow$  signals, valveSt := signals  $\cup$  {open Valve}, openv
   $\square$ 
  closevalve  $\rightarrow$  signals, valveSt := signals  $\cup$  {close Valve}, closev

 $\bullet \mu X \bullet$  startreport  $\rightarrow$  (InitReporter) ; ReportService ; X

```

end

```

channelset TACReporterInterface ==
  { startpumps, stoppumps, openvalve, closevalve, sendprogreedy,
    startreport, reportmode, endreport, failuresrepairs, pumps }
process TACReporter  $\hat{=}$ 
  (TACController
     $\llbracket$  TACReporterInterface  $\rrbracket$ 
    Reporter) \ TACReporterInterface

```

Chapter 6

Steam Boiler

process SteamBoiler $\hat{=}$ TACReporter

Bibliography

- [1] Jean-Raymond Abrial. Steam-Boiler Control Specification Problem. In *Formal Methods for Industrial Applications, Specifying and Programming the Steam Boiler Control (the book grow out of a Dagstuhl Seminar, June 1995)*., pages 500–509, 1995.
- [2] Leonardo Freitas. Circus Example - Parsable Steam Boiler. Technical report, Academia.edu, 2006.
- [3] Jim Woodcock. A Circus Steam Boiler: using the unifying theory of Z and CSP. Technical report, Oxford University Computing Laborator, 2001.
- [4] Kangfeng Ye and Jim Woodcock. Model checking of state-rich formalism *Circus* by linking to *CSP* \parallel *B*. *International Journal on Software Tools for Technology Transfer*, pages 1–24, 2015.