Circus Solution of ESELSystem2 for the ESEL System

Kangfeng Ye

April 27, 2016

1 Header

section *ESELHeader* **parents** *circus_toolkit*

This section gives all basic definitions that will be used in all three *Circus* models. And gateway related definitions are only used in the ESEL System 2.

First of all, three constants are defined. MAX_ESEL and MAX_PID stand for maximum number of displays and maximum number of product categories (or, products for short) in the system separately. And constant $MAX\_GATEWAY$ stands for maximum number of gateways.

$$MAX\_ESEL : \mathbb{N}$$
$$MAX\_PID : \mathbb{N}$$

Then all displays and products are identified by a tag plus a unique number which are defined in the free types $ESID$ and PID below where the constructors ES and PD are the tags for displays and products. For an instance, number ten of the display is given $ES\,10$ or $ES(10)$. Similarly, GID gives all identities for gateways.

$$ESID ::= ES\langle\langle 1 \mathinner{..} MAX\_ESEL\rangle\rangle$$
$$PID ::= PD\langle\langle 1 \mathinner{..} MAX\_PID\rangle\rangle$$

The type of product price is defined as an abbreviation to natural numbers $\mathbb{N}$.

$$Price == \mathbb{N}$$

The unit response is defined as a free type with two constants: uok and $ufail$.

$$UStatus ::= uok \mid ufail$$

The response from this program to the environment is a set of product identities of which the price is not updated successfully due to 1) no linked ESEL ID to the product or 2) failed update to its linked ESEL. The first reason is given the status constant NA and the second is provided the constructor $fail\langle\langle ESID\rangle\rangle$.

$$FStatus ::= fail\langle\langle ESID\rangle\rangle \mid NA$$

Two channels are provided to update the map from ESEL ID to product ID. *updateallmap* will clear all stored map and use the input map as new map, while *updatemap* just updates a partial map. In this map, one ESEL can be linked to up to one product. However, one product may associate with multiple ESELs.

channel $updateallmap : ESID \pfun PID$
channel $updatemap : ESID \pfun PID$

Similarly, two channels are provided to update the price information. *updateallprice* will clear all price information and use the input price information as new price, while *updateprice* just updates price partially.

channel $updateallprice : PID \pfun Price$
channel $updateprice : PID \pfun Price$

The *update* channel gives a signal to the program to start update process.

channel $update$

The *failures* channel returns all failed products and related error reasons after update. Since one product may associate with multiple displays, the return status is a power set of *FStatus* to denote which specific displays that the product links are updated unsuccessfully. But it is worth noting that *NA* and *fail* must not occur in a product's return set at the same time because they can not be both no associate display and associate display update fail.

channel $failures : PID \pfun \mathbb{P}\ FStatus$

The internal *resp* event is used to collect update responses from all displays and *terminate* event is for completing the collection.

channel $resp : PID \times FStatus$
channel $terminate$
channelset $RespInterface == \{\!|\ resp, terminate\ |\!\}$

This *uupdate* event is to update one ESEL to the specific price, and *ures* for update response from this ESEL. And *udisplay* is used to synchronise the show of price on all ESELs at the same time and *finishdisplay* is used to wait for display completion of all ESELs. That is the similar case for *uinit* and *ufinishinit* that are for initialisation synchronisation.

channel $uupdate : ESID \times Price$
channel $ures : ESID \times UStatus$
channel $uinit, finishuinit$
channel $udisplay, finishudisplay$

And *display* is used to synchronise the show of price on all gateways (or ESELs) at the same time and *finishdisplay* is used to wait for display completion of all gateways (or ESELs). That is the similar case for *init* and *finishinit* that are for initialisation synchronisation.

channel $init, finishinit$
channel $display, finishdisplay$

The channels below are for communication between the ESEL system and displays. The *write* event writes price to a display, and the *read* event reads price from the display. *ondisplay* turns on the related display and *offdisplay* turns off it conversely.

channel $write : ESID \times Price$
channel $read : ESID \times Price$
channel $ondisplay : ESID$
channel $offdisplay : ESID$

2 ESEL System 2

section *ESELSystem2* **parents** *ESELHeader*

$$MAX\_GATEWAY : \mathbb{N}$$

$$GID ::= GW\langle\langle 1 \mathinner{..} MAX\_GATEWAY \rangle\rangle$$

The map from ESELs to gateways, *gwmap*, is defined as a total function. One ESEL is linked to up to one gateway. However, one gateway may associate with multiple ESELs.

$$gwmap : ESID \to GID$$
$$gwmap = \{(ES\,1, GW\,1), (ES\,2, GW\,1), (ES\,3, GW\,2)\}$$

The channels below are used to communicate between the server and gateways, or between gateway internals. The server uses *gupdateprice* to send price information with ESEL IDs to the corresponding gateway, while *gfailure* is used to get back the udpate result from the gateway.

channel $gupdateprice : GID \times (ESID \pfun Price)$
channel $gfailure : GID \times \mathbf{P}\ ESID$

gresp and *gterminate* are used in the internal of gateways to collection update results from each ESEL and terminate after collection.

channel $gresp : ESID$
channel $gterminate$
channelset $GRespInterface == \{\!|\, gresp, gterminate \,|\!\}$

ESEL Server Process The process for overall control of the system, named *ESELServer*, is defined as an explicitly defined process.

process $ESELServer \mathrel{\widehat=}$ **begin**

The *ESELServer* has three state components: *pumap* for mapping from ESELs to products, *ppmap* for mapping from products to their price, and *response* for the response of one update to the environment.

$$\begin{aligned}&\textbf{state}\ State == [\, pumap : ESID \pfun PID \,;\, ppmap : PID \pfun Price;\\ &\quad response : PID \pfun (\mathbf{P}\ FStatus)\,]\end{aligned}$$

Initially, these three state components all are empty.

$$Init == [\,(State)' \mid pumap' = \varnothing \land ppmap' = \varnothing \land response' = \varnothing\,]$$

The *UpdateMap* schema updates part of the ESELs to products map according to the input map, while the *UpdateAllMap* schema discards all map and uses new input map as *pumap*.

$$\begin{aligned}&UpdateMap == [\,\Delta State \,;\, map? : ESID \pfun PID \mid\\ &\quad pumap' = pumap \oplus map? \land ppmap' = ppmap \land response' = response\,]\\ &UpdateAllMap == [\,\Delta State \,;\, map? : ESID \pfun PID \mid\\ &\quad pumap' = map? \land ppmap' = ppmap \land response' = response\,]\end{aligned}$$

The *NewPrice* updates part of price information stored, while the *AllNewPrice* discards all price information stored and uses input price as *ppmap*.

$$\begin{aligned}
&NewPrice == [\,\Delta State\,;\, price? : PID \pfun Price \mid \\
&\quad ppmap' = ppmap \oplus price? \land pumap' = pumap \land response' = response\,] \\
&AllNewPrice == [\,\Delta State\,;\, price? : PID \pfun Price \mid \\
&\quad ppmap' = price? \land pumap' = pumap \land response' = response\,]
\end{aligned}$$

AUpdatemap is an action defined to update ESELs to products map: either partial update by *updatemap* event or complete update by *updateallmap* event.

$$\begin{aligned}
&AUpdatemap \mathrel{\widehat=} updatemap?map \then \big(UpdateMap\big) \\
&\quad \Box\ updateallmap?map \then \big(UpdateAllMap\big)
\end{aligned}$$

Similarly, *ANewPrice* is an action defined to update products to price map: either partial update by *updateprice* event or complete update by *updateallprice* event.

$$\begin{aligned}
&ANewPrice \mathrel{\widehat=} updateprice?price \then \big(NewPrice\big) \\
&\quad \Box\ updateallprice?price \then \big(AllNewPrice\big)
\end{aligned}$$

If the update to an ESEL fails, *AUpdateUnitFail* sends the failure by *resp* to the response collection action *CollectResp*.

$$AUpdateUnitFail \mathrel{\widehat=} eid : ESID \bullet resp.(pumap(eid)).(fail\ eid) \then \mathbf{Skip}$$

Or if the product has not been allocated the corresponding ESELs, it sends back a response to state this error *NA*. The behaviour is defined in the *AUpdateNoUnit* action.

$$AUpdateNoUnit \mathrel{\widehat=} pid : PID \bullet resp.pid.NA \then \mathbf{Skip}$$

For all products without associate ESELs, they send the failures independently.

$$\begin{aligned}
&ARespNoUnit \mathrel{\widehat=} \Big|\Big|\Big| pid : (\mathrm{dom}\ ppmap \setminus \mathrm{ran}\ pumap)\ [\![\varnothing]\!] \bullet \\
&\quad AUpdateNoUnit(pid)
\end{aligned}$$

For each gateway, *AUpdateGateways* sends all price for the ESELs which are linked to the gateway and gets back update result. Then for each failure, the action passes it to *AUpdateUnitFail*, and finally writes to *response*.

$$\begin{aligned}
&AUpdateGateway \mathrel{\widehat=} gid : GID \bullet \\
&\quad gupdateprice.gid!((\mathrm{dom}\,(gwmap \rhd \{gid\})) \lhd (pumap \fatsemi ppmap)) \then \\
&\quad gfailure.gid?uids \then \big(\Big|\Big|\Big| uid : uids\ [\![\varnothing]\!] \bullet AUpdateUnitFail(uid)\big)
\end{aligned}$$

Update of price to ESELs is an interleave of *AUpdateGateway* for all gateways.

$$AUpdateGateways \mathrel{\widehat=} \Big|\Big|\Big| gid : GID\ [\![\varnothing]\!] \bullet AUpdateGateway(gid)$$

Then the update of all products, given in the action *AUpdateProducts*, is the interleave of the update of price to ESELs through gateways and the action for the case without associate ESELs. Then it follows a *terminate* event to finish the update.

$$\begin{aligned}
&AUpdateProducts \mathrel{\widehat=} (AUpdateGateways\ [\![\varnothing \mid \varnothing]\!]\ ARespNoUnit)\fatsemi \\
&\quad terminate \then \mathbf{Skip}
\end{aligned}$$

$$AddOneFailure == [\Delta State ; pid? : PID ; fst? : FStatus \mid$$
$$(pid? \in \mathrm{dom}\ response \Rightarrow$$
$$response' = response \oplus \{pid? \mapsto (response(pid?) \cup \{fst?\})\}) \land$$
$$(pid? \notin \mathrm{dom}\ response \Rightarrow$$
$$response' = response \cup \{pid? \mapsto \{fst?\}\}) \land$$
$$ppmap' = ppmap \land pumap' = pumap\,]$$

The *CollectResp* action is to collect responses from all units and write them into the *response* variable. It recursively waits for the response from the units, or terminates if required.

$$ACollectResp \mathrel{\widehat=} \mu X \bullet$$
$$((resp?pid?fst \rightarrow \left(AddOneFailure\right) ; X) \Box terminate \rightarrow \mathbf{Skip})$$

Then update of all products and response collection behaviours are put together into *AUpdateResp* action. It is a parallel composition of *AUpdateProducts* and *CollectResp* actions and they are synchronised with *resp* and *terminate* events. Finally, these internal events are hidden.

$$AUpdateResp \mathrel{\widehat=}$$
$$(AUpdateProducts \,[\![\, \varnothing \mid RespInterface \mid \{response\} \,]\!]\, ACollectResp)$$
$$\setminus RespInterface$$

The overall price update action is given in *AUpdatePrice*, which accepts a *update* event from its environment, then clears *response*, updates the price, sends *display* event to make all ESELs show their price at the same time, then feeds back the response to the environment.

$$AUpdatePrice \mathrel{\widehat=} update \rightarrow response := \varnothing;$$
$$AUpdateResp ; display \rightarrow finishdisplay \rightarrow failures.response \rightarrow \mathbf{Skip}$$

The overall behaviour of the *ESELServer* process is given by its main action. It initializes at first, then repeatedly provides ESEL map update, price map, or price update to its environment.

$$\bullet \left(Init\right) ; init \rightarrow finishinit \rightarrow \mathbf{Skip};$$
$$(\mu X \bullet (AUpdatemap \Box ANewPrice \Box AUpdatePrice) ; X)$$

end

Gateway Process The *Gateway* process is defined as parametrised process.

process $Gateway \mathrel{\widehat=} gid : GID \bullet$ **begin**

It has two state components: *pumap* for the map from ESELs to price, and *failed* for a set of ESELs which update unsuccessfully.

state $State == [\, pumap : ESID \pfun Price ; failed : \mathbf{P}\ ESID\,]$

Initially, both are empty.

$$Init == [\, (State)' \mid pumap' = \varnothing \land failed' = \varnothing\,]$$

The map only can be updated completely and can not be updated partially.

$$UpdateAllMap == [\,\Delta State\ ;\ map? : ESID \pfun Price \mid pumap' = map? \land failed' = failed\,]$$

The map is updated after input from *ESELServer* through the *gupdateprice* channel.

$$AUpdateallmap \mathrel{\widehat=} gupdateprice.gid?map \then \big(UpdateAllMap\big)$$

A parameterised action, *AUpdateUnitPrice*, is given to update the price (specified by the formal *pid* parameter) to an ESEL (given by the formal *uid* parameter). It sends the price to the specified ESEL by the *uupdate* event, and then waits for the response from the ESEL. If the return status is not successful (*ufail*), it sends the result to response collection action *CollectResp* below, then terminates. Otherwise, it terminates immediately.

$$\begin{aligned}&AUpdateUnitPrice \mathrel{\widehat=} uid : ESID \bullet\\ &\quad uupdate.uid.(pumap\ uid) \then ures.uid?rst \then\\ &\quad \big((rst = ufail)\ \&\ gresp!uid \then \mathbf{Skip}\\ &\quad \Box\ (rst = uok)\ \&\ \mathbf{Skip}\big)\end{aligned}$$

Updates of all ESELs in this gateway are put in an iterated interleave, then follow a *gterminate* event to finish the updates.

$$\begin{aligned}&AUpdateAllUnits \mathrel{\widehat=} ((\Big|\Big|\Big|\ eid : (\mathrm{dom}\ pumap)\ @[\![\ \varnothing]\!]\ \bullet\ AUpdateUnitPrice(eid))\\ &\quad ;gterminate \then \mathbf{Skip})\end{aligned}$$

The *CollectResp* action is to collect responses from all units and write them into the *response* variable. It recursively waits for the response from the units, or terminates if required.

$$\begin{aligned}&AGCollectResp \mathrel{\widehat=} \mu X \bullet\\ &\quad ((gresp?uid \then failed := failed \cup \{uid\}\ ;\ X)\ \Box\ gterminate \then \mathbf{Skip})\end{aligned}$$

Then update of all products and response collection behaviours are put together into *AUpdateResp* action. It is a parallel composition of *AUpdateProducts* and *CollectResp* actions and they are synchronised with *resp* and *terminate* events. Finally, these internal events are hidden.

$$\begin{aligned}&AGUpdateResp \mathrel{\widehat=}\\ &\quad (AUpdateAllUnits\ [\![\ \varnothing \mid GRespInterface \mid \{failed\}\ ]\!]\ AGCollectResp)\\ &\quad \setminus GRespInterface\end{aligned}$$

The overall price update action is given in *AUpdatePrice*, which accepts a *gupdateprice* event from its environment, then clears *failed*, updates the price, sends update results to the server, and waits for *display* event to make all ESELs in this gateway show their price at the same time.

$$\begin{aligned}&AGUpdatePrice \mathrel{\widehat=} AUpdateallmap\ ;\ failed := \varnothing;\\ &\quad AGUpdateResp\ ;\ gfailure.gid!failed \then display \then udisplay \then\\ &\quad finishudisplay \then finishdisplay \then \mathbf{Skip}\end{aligned}$$

The overall behaviour of the *Gateway* process is given by its main action. It initializes at first, then repeatedly provides ESEL map update, price map, or price update to its environment.

$$\bullet \left(Init\right) ; init \rightarrow uinit \rightarrow finishuinit \rightarrow finishinit \rightarrow \mathbf{Skip};$$
$$(\mu X \bullet (AGUpdatePrice) ; X)$$

end

ESEL Process Each ESEL is defined as a parameterised process with the formal parameter—ESEL ID.

$$\mathbf{process}\ ESEL2 \mathrel{\widehat=} eid : ESID \bullet \mathbf{begin}$$

The process has two state components: *price* for the price to display, and *status* for the status of ESEL.

$$\mathbf{state}\ State == [\, price : Price \,;\, status : UStatus \,]$$

Initially, the price is equal to 0 and the status is *uok*.

$$Init == [\, (State)' \mid price' = 0 \land status' = uok \,]$$

The *Update* action provides its environment (*Gateway*) the update of price for the associated product. It accepts the *uupdate* event with the price, then writes the price to *price*. After that, it writes the price to the display unit, and reads back the value to compare with the original price. If it is equal, it sends back status *uok* by the *ures* event. Otherwise, it sends back status *ufail*. Accordingly, *status* is updated.

$$Update \mathrel{\widehat=} uupdate.eid?x \rightarrow price := x \,;\, write.eid.price \rightarrow read.eid?y$$
$$\rightarrow ((y = price) \,\&\, ures.eid.uok \rightarrow status := uok$$
$$\Box\, (y \neq price) \,\&\, ures.eid.ufail \rightarrow status := ufail)$$

The *Display* action accepts the *udisplay* event. If the status is *uok*, then the associated display is turned on. Otherwise, the display is turned off.

$$Display \mathrel{\widehat=} udisplay \rightarrow (\ (status = uok) \,\&\, ondisplay.eid \rightarrow \mathbf{Skip}$$
$$\Box\, (status = ufail) \,\&\, offdisplay.eid \rightarrow \mathbf{Skip})$$
$$;finishudisplay \rightarrow \mathbf{Skip}$$

$$NotUpdateDisplay \mathrel{\widehat=} udisplay \rightarrow offdisplay.eid \rightarrow finishudisplay \rightarrow \mathbf{Skip}$$

The initial behaviour of the process is given in the action *AInit* which initialises the state at first, and then turns off the display.

$$AInit \mathrel{\widehat=} \left(Init\right) ; uinit \rightarrow offdisplay.eid \rightarrow finishuinit \rightarrow \mathbf{Skip}$$

The overall behaviour of the process is given by its main action. It specifies that after initialisation the process repeatedly provides update or display to its environment.

$$\bullet\ AInit\ ;\ (\mu X \bullet ((Update\ ;\ Display)\ \Box\ NotUpdateDisplay)\ ;\ X)$$

end

System All ESELS which are registered with the same gateway synchronise on unit initialisation and display events.

channelset $InterESELInterface2 == \{\!|\ uinit, finishuinit, udisplay, finishudisplay\ |\!\}$
process $ESELS2 \mathrel{\widehat=} gid : GID \bullet$
$(\|\ eid : (\mathrm{dom}\,(gwmap \rhd \{gid\}))\ [\![\ InterESELInterface2]\!]\ \bullet\ ESEL2(eid))$

Each gateway is in parallel with its linked ESELs in the *GatewayESELS* process. And all gateays synchronise on gateway initialisation and display events which is defined as the *Gateways* process.

channelset $InterGWInterface2 == \{\!|\ init, finishinit, display, finishdisplay\ |\!\}$
channelset $GWESELInterface2 == \{\!|\ uinit, finishuinit, uupdate, ures, udisplay, finishudisplay\ |\!\}$
process $GatewayESELS \mathrel{\widehat=} gid : GID \bullet$
$(Gateway(gid)\ [\![\ GWESELInterface2\ ]\!]\ ESELS2(gid)) \setminus GWESELInterface2$
process $Gateways \mathrel{\widehat=} \|\ gid : GID\ [\![\ InterGWInterface2]\!]\ \bullet$
$GatewayESELS(gid)$

Finally, the ESEL System 2 is simply the parallel composition of the *ESELServer* and the *Gateways*, and communications between them are hidden.

channelset $ServerGWInterface == \{\!|\ init, finishinit, gupdateprice, gfailure, display, finishdisplay\ |\!\}$
process $ESELSystem2 \mathrel{\widehat=}$
$(ESELServer\ [\![\ ServerGWInterface\ ]\!]\ Gateways) \setminus ServerGWInterface$